// File: src/SchruteBucks.sol



// SPDX-License-Identifier: MIT

// ooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooooollllloooooooooooooooooooooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooooooooooooooooooooooooooollllccccccllllllccclloooooooooooooooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooooooooooooooooooolc:;,'...............',;:ccccclooooooooooooooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooooooooooooooolc:,'...                      .';:::loooooooooooooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooooooooooool:,.........                        ...',,:clooooooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooooooool:,..........                                 ..,:loooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooooool;............                                     .':loooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooool;.....''..........                                   ..,cooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooool;....',,,,'''..........                                ....;looooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooo:.....',;;,,,,,,''.........          ........',;:;.      .....,coooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooool,. ....',,,,;;;,,,,,''.....    ....',:clooodddxkOOOd;.    ......'cooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooo:.  ......''',,''',;,,,,,'.. ..';coddkO000K0000000000Ox:. ....'''..'coooooooooooooooooooooooooooooo
// ooooooooooooooooooooooo;.  ..............'''',,,....,cdkO0KKKKKKKKKKKK00000000kd:.....',,'..'cooooooooooooooooooooooooooooo
// oooooooooooooooooooooo;.   .................''...'cdO0KXXXXXXXXKKKKKKKKK000000Okd:....'''....'coooooooooooooooooooooooooooo
// ooooooooooooooooooooo:.       ......   ........'cx0KXXXXXXXXXXXXKKKKKKKK0000000Oxo;...........,looooooooooooooooooooooooooo
// ooooooooooooooooooooc.          ..      ... ..ck0XXXXXXXXXXXXXXXXKKKKKKKK0000000kdl'......... .:ooooooooooooooooooooooooooo
// oooooooooooooooooool,                      .:x0XXXXXXXXXXXXXXXXXXXKKKKKKKK000000Oxo:. ......  .'loooooooooooooooooooooooooo
// oooooooooooooooooooc.              ..    .,oOKXXXXXNXXXXXXXXXXXXXXKKKKKKKK0000000kol;.         .;oooooooooooooooooooooooooo
// ooooooooooooooooooo:.             .'.   .cx0XXXXXXXNNNNXXXXXXXXXXKKKKKKKKK0000000Odoc,          'looooooooooooooooooooooooo
// ooooooooooooooooooo:.             ',. .,ok0XXXXXNNNNNXXXXXXXXXXXXKKKKKKKKK0000000Oxolc.         .:ooooooooooooooooooooooooo
// ooooooooooooooooooo:.            .;' .;dOKXXXXXNNNNXXXXXXXXXXXXXXKKKKKKKKK0000000Okdll:.         ,loooooooooooooooooooooooo
// ooooooooooooooooooo:.            ,:..:xOKXXXXXXXXNNNNXXXXXXXXXXXXKKKKKKKKK00000000kdool'         'loooooooooooooooooooooooo
// oooooooooooooooooooc.           .c;.;xO0KXXXXXXXXNNXXXXXXXXXXXXXXKKKKKKKKK00000000Odool;.        .coooooooooooooooooooooooo
// ooooooooooooooooooo:.           ,l;;dO0KXXXXXXXXXXXXXXXXXXXXXXXXKKKKKKKKK000000000Oxool:'        .:oooooooooooooooooooooooo
// ooooooooooooooooooo:.          .:ocoOOOOkxxkO0KXXXXXXXXXXXXXXKKKKKKKKKKKK0000000Okdolll:,.       'coooooooooooooooooooooooo
// ooooooooooooooooooo;.          .lddkOxl:;,''',;:loxO0KKKKKKK00KKKK000KKK00Okxol:,'...,;;,.       'coooooooooooooooooooooooo
// ooooooooooooooooool,           .lxxkkxdddolc:;'......,;:loxkdx0K00kdxxoc:;'.......',,;;;,.       'loooooooooooooooooooooooo
// ooooooooooooooooooo:. ...      .':loc;;ccccccc:,'..     .';:cx0OOkd:'...    ...',''''.....       'loooooooooooooooooooooooo
// oooooooooooooooooool'.,:.      .;,'...,;:ccclcc:;;,,'....';cdO0Okxd:.......'',,;;;'.... .        .coooooooooooooooooooooooo
// oooooooooooooooooool'.:o,..    ,xkd,'cc;,':c,...';;,,,,;::,.,ldoc:'...'''''..';:;,.......        'coooooooooooooooooooooooo
// oooooooooooooooooool,.:dl,,.   ,xOO:,xOkocdk:...;dxc;;:ldx:...;;,.  .,,:l;'..,lOk:....',.   .....,loooooooooooooooooooooooo
// oooooooooooooooooooo;.:xxoc,''.'oOOl;d00kl:lc;,;okxoloxOOo'.'cool;..,lxxxo;.'cxOo,.,,';,....;;;cccloooooooooooooooooooooooo
// oooooooooooooooooooo:.,xOkoclo:;okOx;:k00kdc::::clodxO0kl,,:ldxxd:..,dOOxo:,;:c;,,;c:;;. .;cooooooooooooooooooooooooooooooo
// ooooooooooooooooooool'.lkOkddkxodxkOxc:codxkkkkkkkkkxdl;,cxOO0000d:,':dOOOxddoollllc;''...coooooooooooooooooooooooooooooooo
// ooooooooooooooooooooo:.'lkOkkkxxxxkOOOxocc::::c::::;;:coxO000OOOOxc:c:;:cclooolc:;;,',;'.'loooooooooooooooooooooooooooooooo
// oooooooooooooooooooooo:'.,:::cdxkkkkO0000OOkxxxxxxxkO00KK0OOOOOOOxl;cxxdlc:::::;;:::::;..,loooooooooooooooooooooooooooooooo
// oooooooooooooooooooooool;.   .cxkkkkOOO0000KKKKKKKKKKKK0OOOOOO000ko;,oOOOOOkkkxxxddolc;..;ooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooool'   .:dkkOOOOOOO0000KKKKKKKKK0kkxollldxxo:'.:xOOOOOOOOkkkkxoc,..cooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooool'    ;dkOOOOOOOOO000000KKKKKK0xl:;;,;cc;....,okOOOOkkkkkxxdl:'.'looooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooool,    'okkOOOOOOOOO0000000KKKKK0xdxxxoc;'...,cdkOOOkkkxxddool:..;oooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooo:,.  .cxkOOOOOOOOOOO0000KKKKKKKK00OOxollccldkOOOOOkkxddooolc;..:oooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooc,..,okkOOOOOOOOOOO000KKKKKK000000OkOOkkkkkOOOOOkxxdooool:'.'loooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooolc'.;dkkOOOOOOOOOOO000000000OkkkkkxxxxxkkkkkOOkkxdoloolc;..:ooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooooc'.;oxkkOOOOOOOOO0Okdloddolllllcccccccllcoxkkxdollllc;..;oooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooooool,.'ldxkkOOOkkOOOkxdolloodxxkxxxxddolc:coxkxdolccc:;..,loooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooooooo:..;odxkkkkkkkkkOOOOOOOkxdddolclodddxxkkxdolc::;,..;looooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooooooooo:..':odxkkxxxxxkOOOOkxdolllcc::clodxxxxdoc:;;,...:ooooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooooooooc..'.';ldxxxxxxkkkkkkkkkOOOOOkkxddxxxddlc;;,'..;loooooooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooooooooooool;..cc,..,codxxxkkkkOOO00000000Okxxxxdol:;,'. .;looooooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooooool:'.. .cddl;...,:ldxxkkOOOOOOOOOOOkxdddolc;,'.    ..;looooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooooool;..... .cddxdoc,...';coddxxxxxxdddddoolc:;'.. ..      .,cooooooooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooooool:'...,,. .:dxxxkkxoc;'...',;::::::::;;,,'...  .,,.   .    .':ooooooooooooooooooooooooooooooooooo
// oooooooooooooooooooooooool;....'c;  .:xkkkkkkkkxxol;'............       .::,.  .'.     .':loooooooooooooooooooooooooooooooo
// ooooooooooooooooooooooool;'....:dl...;dkkkkkOkkkkOOOOkoc;..            .:c:' .';;.        .;loooooooooooooooooooooooooooooo
// oooooooooooooooooooooooc;'''..'d0kc..,dkkOOOOOOOOkOOO0KK0Oxdl:'.      .;::'..:c:;'.         .';:llooooooooooooooooooooooooo
// oooooooooooooooooolc:;,''''...:kKXOl..cxkOOOOOOOOOOkkkO0KKK00Od:... .';c:'.;lol:;'.             ..',:cloooooooooooooooooooo
// oooooooooooollc;,'.....'''.. .oOKNX0d;'cdkOOOOOOOOOOOkxxkO00Oko:'...,cl:,:dkxoc:;,.                  ..',;:lloooooooooooooo
// ooooolcc:;,'...    ..'''''.  'dO0XNNXOc,;lxOOOOOOOOOOOkxdodxdo:'',,;ll:cd00kdl:;;,......                   ..',;:cllooooooo
// c;,'....        ...''''''....'oOOKNNNNKx:,;okOOOOOOOOOOkxolcc;,,;:ccccd0K0Oxl:;;;,.............                  ....',;:cl
// ,.           ...''''''''.. .''lOOOKNWWNX0o;,:dkOOOOOOOOOkxolc::cccclx0XK0Oxlc:::;. ...............   .........          ...
// ,.............''''''''''. .::':k0OOKXNNWNXOo;,:dkOOOOOOOOkddollclokKXXXK0xlcccc:'  ................  ................      
// ;...........'''''''''''.  ,dd'.oO00O0XNNWNNXOo;,;lxkOOOOkxoolcldOKXXXXX0xollooc'..................... .....................
// :'''''....'''''''''''... .l00l.,d0KK000XNNWWNX0d:,,:ldddlcllokKXNNNNXKOxoloddo,...................... .....................
// :'''....'''''''''''....  'xXXO:.,x0KXXK00KXNNNNXOo'.','.,cxk0XNNNNXK0kdooxkkd;..;'.................... ....................
// :''....''''''''''......  ;x0KKk:.,dOKNNNXK0KXXXKx;........;coOKXXK0OkxddkOkd;..::...................... ...................
// :'...'''''''''''......  .o0000Okc.'oOKXNNNNXK0kc..:cccllc,...,oO0000OxkOOOd;.'od;......................  ..................
// :....'''''''''........  ,ONNNXXK0o''lk0KNNNNKo'.':ldxkkxdl,.. .:xO00OO00Od;.,xKx'.......................  .................
// ;.....''''''..........  :0NNNNNNNXx,.lk0XNXk:. .lxkkkxddo:.   ..'ok0000Od;.;kXKl......................... .................
// ;.......'''......'''.. .cKNNNNNNNNXk,.ck00o'';..;xOOxxddc.   'c;..cxO0Oo,.:kXNk,.''.....................   ................
// :''..........'''''''.. .oKXXXXNNNNNNk;'cd:':k0o..ckkxxdl'  .,okOd'.:xko,.cOXNKl..''.......  .........     .................
// :'''......''',,'''''.  'xXNNNXXXNNNNNO:'',lO0K0l.'dxddo;  ,coOXX0d,.;c,'lOXNNk,.''.......... .....     ....................
// :''.....'',,',,''''.. .:OKXXXXXXNNNNNN0xdkOO000o..'',c:. .cdkKXX0kd:'';oONNNKl.'''............     ........................
// :'....',,,,,,,'''''.  .oKKKKKKKKKXXXNNNNNNNNXXO;.;,.'oo' .;odOKK000klcoOXNNNO;.''..............   .........................

pragma solidity ^0.8.7;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";

contract SBUCKS is ERC20, Ownable {
    constructor() ERC20("SBUCKS", "SBUCKS"){}

    /// turn on/off contributions
    bool public allowContributions;

    /// a minimum contribution to participate in the presale
    uint256 public constant MIN_CONTRIBUTION = .025 ether;

    /// limit the maximum contribution for each wallet
    uint256 public constant MAX_CONTRIBUTION = .35 ether;

    /// the maximum amount of eth that this contract will accept for presale
    uint256 public constant HARD_CAP = 17.5 ether;

    /// total number of tokens available
    uint256 public constant MAX_SUPPLY = 696969696969 * 10 ** 18;

    /// 40% of tokens reserved for presale
    uint256 public constant PRESALE_SUPPLY = 331060606060 * 10 ** 18;

    /// 60% for LP and team
    uint256 public constant RESERVE_MAX_SUPPLY = 365909090909 * 10 ** 18;

    /// used to track the total contributions for the presale
    uint256 public TOTAL_CONTRIBUTED;

    /// used to track the total number of contributoors
    uint256 public NUMBER_OF_CONTRIBUTOORS;

    /// mapping of whitelist addresses
    mapping(address => bool) public whiteLists;

    /// mapping of blacklist addresses
    mapping(address => bool) public blackLists;

    /// boolean for public sale
    bool public publicSale = false;

    /// boolean for limiting holdings initially
    bool public limited;

    /// set max holding amount
    uint256 public maxHoldingAmount;

    /// uniswapV2Pair address
    address public uniswapV2Pair;

    /// a struct used to keep track of each contributoors address and contribution amount
    struct Contribution {
        address addr;
        uint256 amount;
    }

    function setRule(bool _limited, uint256 _maxHoldingAmount) external onlyOwner {
        limited = _limited;
        maxHoldingAmount = _maxHoldingAmount;
    }

    function whiteList(address _address, bool _isWhiteListing) external onlyOwner{
        whiteLists[_address] = _isWhiteListing;
    }

    function blackList(address _address, bool _isBlackListing) external onlyOwner{
        blackLists[_address] = _isBlackListing;
    }

    function addBatchWhiteList(address[] calldata _address) external onlyOwner {
        for(uint i = 0; i < _address.length; i++) {
            whiteLists[_address[i]] = true;
        }
    }

    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) override internal virtual {
        require(!blackLists[to] && !blackLists[from], "Blacklisted");

        if (uniswapV2Pair == address(0) && from != address(0)) {
            require(from == owner(), "trading is not started");
            return;
        }

        if(limited && from == uniswapV2Pair) {
            require(super.balanceOf(to) + amount <= maxHoldingAmount, "Forbid");
        }
    }

    /// mapping of contributions
    mapping (uint256 => Contribution) public contribution;

    /// index of an address to it's contribition information
    mapping (address => uint256) public contributoor;

    /// collect presale contributions
    function sendToPresale() public payable {
        if(!publicSale) {
            require(whiteLists[msg.sender], "You are not whitelisted.");
        }

        /// look up the sender's current contribution amount in the mapping
        uint256 currentContribution = contribution[contributoor[msg.sender]].amount;

        /// initialize a contribution index so we can keep track of this address' contributions
        uint256 contributionIndex;

        require(msg.value >= MIN_CONTRIBUTION, "Contribution too low");

        /// check to see if contributions are allowed
        require (allowContributions, "Contributions not allowed");

        /// enforce per-wallet contribution limit
        require (msg.value + currentContribution <= MAX_CONTRIBUTION, "Contribution exceeds per wallet limit");

        /// enforce hard cap
        require (msg.value + TOTAL_CONTRIBUTED <= HARD_CAP, "Contribution exceeds hard cap"); 

        if (contributoor[msg.sender] != 0){
            /// no need to increase the number of contributors since this person already added
            contributionIndex = contributoor[msg.sender];
        } else {
            /// keep track of each new contributor with a unique index
            contributionIndex = NUMBER_OF_CONTRIBUTOORS + 1;
            NUMBER_OF_CONTRIBUTOORS++;
        }

        /// add the contribution to the amount contributed
        TOTAL_CONTRIBUTED = TOTAL_CONTRIBUTED + msg.value;

        /// keep track of the address' contributions so far
        contributoor[msg.sender] = contributionIndex;
        contribution[contributionIndex].addr = msg.sender;
        contribution[contributionIndex].amount += msg.value;
    }

    function airdropPresale() external onlyOwner {

        /// determine the price per token
        uint256 pricePerToken = (TOTAL_CONTRIBUTED * 10 ** 18)/PRESALE_SUPPLY;

        /// loop over each contribution and distribute tokens
        for (uint256 i = 1; i <= NUMBER_OF_CONTRIBUTOORS; i++) {

            /// convert contribution to 18 decimals
            uint256 contributionAmount = contribution[i].amount * 10 ** 18;

            /// calculate the percentage of the pool based on the address' contribution
            uint256 numberOfTokensToMint = contributionAmount/pricePerToken;

            /// mint the tokens to the address
            _mint(contribution[i].addr, numberOfTokensToMint);
        }

        /// calculate the remaining supply
        uint256 numberToMint = MAX_SUPPLY - totalSupply();
        
        /// don't allow the dev mint until the tokens have been airdropped
        require (numberToMint <= RESERVE_MAX_SUPPLY, "Dev mint limited to reserve max");

        /// mint the remaining supply to the dev's wallet
        _mint(msg.sender, numberToMint);
    }

    /// set whether or not the contract allows contributions
    function setAllowContributions(bool _value) external onlyOwner {
        allowContributions = _value;
    }

    /// if there are not enough contributions or we decide this sucks, refund everyone their eth
    function refundEveryone() external onlyOwner {
        for (uint256 i = 1; i <= NUMBER_OF_CONTRIBUTOORS; i++) {
            address payable refundAddress = payable(contribution[i].addr);
            
            /// refund the contribution
            refundAddress.transfer(contribution[i].amount);
        }
    }

    /// allows the owner to withdraw the funds in this contract
    function withdrawBalance(address payable _address) external onlyOwner {
        (bool success, ) = _address.call{value: address(this).balance}("");
        require(success, "Withdraw failed");
    }

    /// toggle public sale
    function togglePublic() external onlyOwner() {
        publicSale = !publicSale;
    }

    /// allows the owner to set uniswapv2pair
    function setUniswapV2Pair(address _uniswapV2Pair) external onlyOwner {
        uniswapV2Pair = _uniswapV2Pair;
    }
}
// File: lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}